

ATIVIDADE EXTRA-CLASSE

3 – Estrutura Condicional GABARITO

A-) Escreva algoritmos para solução dos problemas a seguir:

1-) Leia dois números inteiros e imprima “***Os números são iguais!***”, caso eles sejam iguais.

```
algoritmo "Ex01"
var
numero1, numero2: inteiro
inicio
    escreva("Digite o número 1:")
    leia(numero1)
    escreva("Digite o número 2:")
    leia(numero2)
    se (numero1=numero2) entao
        escreval("Os números são iguais!")
    fimse
finalgoritmo
```

2-) Leia dois números inteiros e imprima “***Os números são iguais!***”, caso eles sejam iguais ou a mensagem “***Os números são diferentes!***”, caso contrário.

```
algoritmo "Ex02"
var
numero1, numero2: inteiro
inicio
    escreva("Digite o número 1:")
    leia(numero1)
    escreva("Digite o número 2:")
    leia(numero2)
    se (numero1=numero2) entao
        escreval("Os números são iguais!")
    senao
        escreva("Os números são diferentes!")
    fimse
finalgoritmo
```

3-) Leia dois números inteiros e imprima “***Os números são iguais!***”, caso eles sejam iguais ou caso eles sejam diferentes, verifique ainda qual deles é o maior, imprimindo na tela “***## é maior que ##***” ou “***## é menor que ##***”.

```
algoritmo "Ex03"
var
numero1, numero2: inteiro
inicio
    escreva("Digite o número 1:")
    leia(numero1)
    escreva("Digite o número 2:")
    leia(numero2)
    se (numero1=numero2) entao
        escreval("Os números são iguais!")
    senao
        se (numero1>numero2) entao
            escreva(numero1, " é maior que ", numero2)
        senao
            escreva(numero1, " é menor que ", numero2)
        fimse
    fimse
finalgoritmo
```

4-) Leia dois nomes e imprima "**Nomes iguais: #####**", caso eles sejam iguais.

```
algoritmo "Ex04"
var
nome1, nome2: caractere
inicio
    escreva("Digite o nome 1: ")
    leia(nome1)
    escreva("Digite o nome 2: ")
    leia(nome2)
    se (nome1=nome2) entao
        escreva("Nomes iguais: ", nome1)
    fimse
finalgoritmo
```

5-) Leia dois nomes e imprima "**Nomes iguais: #####**", caso eles sejam iguais ou "**Nomes diferentes: #### e ####**", caso contrário.

```
algoritmo "Ex05"
var
nome1, nome2: caractere
inicio
    escreva("Digite o nome 1: ")
    leia(nome1)
    escreva("Digite o nome 2: ")
    leia(nome2)
    se (nome1=nome2) entao
        escreva("Nomes iguais: ", nome1)
```

```
senao
    escreva("Nomes diferentes: ", nome1, " e ", nome2)
fimse
finalgoritmo
```

6-) Leia 3 números inteiros e imprima a mensagem “## é o maior número”, verificando dentre os 3 qual é o maior valor.

```
algoritmo "Ex06"
var
numero1, numero2, numero3: inteiro
inicio
    escreva("Digite o número 1:")
    leia(numero1)
    escreva("Digite o número 2:")
    leia(numero2)
    escreva("Digite o número 3:")
    leia(numero3)
    se (numero1>numero2) entao
        se (numero1>numero3) entao
            escreva(numero1, " é o maior número")
        senao
            escreva(numero3, " é o maior número")
        fimse
    senao
        se (numero2>numero3) entao
            escreva(numero2, " é o maior número")
        senao
            escreva(numero3, " é o maior número")
        fimse
    fimse
finalgoritmo
```

7-) Leia o valor do litro do Etanol e o valor do litro da Gasolina e mostre a mensagem “***Etanol é mais vantajoso do que Gasolina!***” ou vice-versa. A fórmula a seguir indica isso:

$$\text{valor_etanol} / \text{valor_gasolina} \leq 70\%$$

```
algoritmo "Ex07"
var
vl_etanol, vl_gasolina: real
inicio
    escreva("Digite o preço do litro do Etanol:")
    leia(vl_etanol)
    escreva("Digite o preço do litro da Gasolina:")
    leia(vl_gasolina)
    se (vl_etanol / vl_gasolina <= 0.7) entao
```

```
    escreval("Etanol é mais vantajoso do que Gasolina!")
senao
    escreval("Gasolina é mais vantajoso do que Etanol!")
fimse
fimalgoritmo
```

8-) Elaborar um algoritmo que leia 2 valores a e b e os escreva com a mensagem: “**São múltiplos**” ou “**Não são múltiplos**” (b/a não possui resto, então eles são divisíveis).

```
algoritmo "Ex08"
var
a, b: inteiro
inicio
    escreva("Digite o valor de a:")
    leia(a)
    escreva("Digite o valor de b:")
    leia(b)
    se (b mod a <> 0) entao
        escreva(a, " e ", b, " não são Múltiplos!")
    senao
        escreva(a, " e ", b, " são Múltiplos!")
    fimse
fimalgoritmo
```

9-) Um usuário deseja um algoritmo onde possa escolher que tipo de média deseja calcular a partir de 2 notas. Faça um algoritmo que leia as notas, a opção escolhida pelo usuário e calcule a média.

Opção	Tipo de Média
1	aritmética normal
2	ponderada (peso 4 e 6)

```
algoritmo "Ex09"
var
nota1, nota2, media: real
opcao: inteiro
inicio
    escreva("Digite a nota 1: ")
    leia(nota1)
    escreva("Digite a nota 2: ")
    leia(nota2)
    escreva("Escolha Média (1-Normal / 2-Ponderada): ")
    leia(opcao)
    se (opcao = 1) entao
        media <- (nota1 + nota2) / 2
        escreva("Média normal = ", media)
```

```
senao
    media <- (nota1*4 + nota2*6) / 10
    escreva("Média ponderada = ", media)
fimse
fimalgoritmo
```

10-) Um vendedor necessita de um algoritmo que calcule o preço total devido por um cliente. O algoritmo deve receber o código de um produto e a quantidade comprada e calcular o preço total, mostrando-o na tela; use a tabela abaixo:

Código do Produto	Preço unitário
1001	5,32
1324	6,45
6548	2,37

Caso o código do produto não seja válido, mostrar a mensagem: "Código Inválido!".

```
algoritmo "Ex10"
var
total: real
codigo, quantidade: inteiro
inicio
    escreva("Digite o código o produto: ")
    leia(codigo)
    escreva("Digite a quantidade: ")
    leia(quantidade)
    se (codigo = 1001) entao
        total<-quantidade * 5.32
    senao
        se (codigo = 1324) entao
            total<-quantidade * 6.45
        senao
            se (codigo = 6548) entao
                total <- quantidade * 2.37
            senao
                total <- 0
            fimse
        fimse
    fimse
    se (total > 0) entao
        escreva("Total = ", total)
    senao
        escreva("Código Inválido!")
    fimse
fimalgoritmo
```

11-) Escreva um algoritmo que leia o RA, nome de um aluno e suas 2 notas. Calcule a média ponderada do aluno, considerando que o peso para a primeira nota seja 4 e para a segunda nota seja 6. Mostre o RA do aluno, seu nome, suas notas, a média calculada e uma mensagem "**APROVADO**" se a média for maior ou igual a 5 e "**REPROVADO**" se a média for menor que 5. Além disso, caso sua nota seja superior ou igual a 9.0, apareça a mensagem "**APROVADO COM LOUVOR!**".

```
algoritmo "Ex11"
var
ra, nome: caractere
nota1, nota2, media: real
inicio
    escreva("Digite o RA:")
    leia(ra)
    escreva("Digite o nome:")
    leia(nome)
    escreva("Digite a nota 1:")
    leia(nota1)
    escreva("Digite a nota 2:")
    leia(nota2)
    media <- (nota1 * 4 + nota2 * 6) / 10
    escreval("RA Aluno: ", ra)
    escreval("Nome: ", nome)
    escreval("Nota 1: ", nota1)
    escreval("Nota 2: ", nota2)
    escreval("Média: ", media)
    se (media >= 5.0) entao
        se (media >= 9.0) entao
            escreval("APROVADO COM LOUVOR!")
        senao
            escreval("APROVADO!")
        fimse
    senao
        escreval("REPROVADO!")
    fimse
finalgoritmo
```

12-) Modifique o exemplo anterior de forma que seja considerada a frequência do mesmo e a classificação de sua situação de acordo com a tabela a seguir.

Média	Frequência	Situação
>= 5.0	>= 70%	Aprovado
	< 70%	RDR
< 5.0	>= 70%	RDR
	< 70%	Reprovado

```
algoritmo "Ex12"
var
ra, nome: caractere
nota1, nota2, media, frequencia: real
inicio
    escreva("Digite o RA:")
    leia(ra)
    escreva("Digite o nome:")
    leia(nome)
    escreva("Digite a nota 1:")
    leia(nota1)
    escreva("Digite a nota 2:")
    leia(nota2)
    escreva("Digite a frequência:")
    leia(frequencia)
    media <- (nota1 * 4 + nota2 * 6) / 10
    escreval("RA Aluno: ", ra)
    escreval("Nome: ", nome)
    escreval("Nota 1: ", nota1)
    escreval("Nota 2: ", nota2)
    escreval("Média: ", media)
    escreval("Frequência: ", frequencia)
    se (media >= 5.0) entao
        se (frequencia >= 70) entao
            escreva("APROVADO")
        senao
            escreva("RDR")
        fimse
    senao
        se (frequencia >= 70) entao
            escreva("RDR")
        senao
            escreva("REPROVADO")
        fimse
    fimse
finalgoritmo
```

13-) Faça um algoritmo que leia um nº inteiro e mostre uma mensagem indicando se este número é par ou ímpar, e também positivo ou negativo.

```
algoritmo "Ex13"
var
numero: inteiro
inicio
    escreva("Digite um número: ")
    leia(numero)
    se ((numero mod 2) = 0) entao
        escreva("NÚMERO PAR ")
    senao
```

```
    escreva("NÚMERO ÍMPAR ")
fimse
se (numero >= 0) entao
    escreva("E POSITIVO!")
senao
    escreva("E NEGATIVO!")
fimse
fimalgoritmo
```

14-) O cardápio de uma lanchonete é o seguinte:

Especificação	Código	Preço
Cachorro quente	100	2,50
Bauru simples	101	3,00
Cheeseburger	102	3,50
X-Salada	103	3,00

Escrever um algoritmo que leia o código do item pedido, a quantidade e calcule o valor a ser pago por aquele lanche.

```
algoritmo "Ex14"
var
codigo, quantidade: inteiro
total_pedido, preco: real
inicio
    escreva("Digite o código: ")
    leia(codigo)
    escreva("Digite a quantidade: ")
    leia(quantidade)
    preco <- 0
    se (codigo = 100) entao
        escreval("Cachorro quente")
        preco <- 2.5
    fimse;
    se (codigo = 101) entao
        escreval("Bauru simples")
        preco <- 3.0
    fimse
    se (codigo = 102) entao
        escreval("Cheeseburger")
        preco <- 3.5
    fimse
    se (codigo = 103) entao
        escreval("X-Salada")
        preco <- 3.0
    fimse
    total_pedido <- quantidade * preco
    escreval("Total do Pedido = ", total_pedido)
fimalgoritmo
```


15-) Tendo como dados de entrada a altura e o sexo de uma pessoa ("M" masculino e "F" feminino), construa um algoritmo que calcule seu peso ideal, utilizando as seguintes fórmulas:

- para homens: $(72.7 \cdot h) - 58$
- para mulheres: $(62.1 \cdot h) - 44.7$

```
algoritmo "Ex15"
var
sexo: caractere
altura, peso_ideal: real
inicio
    escreva("Digite o sexo: ")
    leia(sexo)
    escreva("Digite a altura: ")
    leia(altura)
    se (sexo = "M") entao
        peso_ideal <- (72.7 * altura) - 58
    senao
        peso_ideal <- (62.1 * altura) - 44.7
    fimse
    escreval("Peso ideal = ", peso_ideal)
fimalgoritmo
```

16-) Modifique o exercício anterior de forma que seja perguntado o peso da pessoa e verifique se o mesmo está dentro ou fora do peso ideal, mostrando uma mensagem correspondente.

```
algoritmo "Ex16"
var
sexo: caractere
altura, peso_ideal, peso: real
inicio
    escreva("Digite o sexo: ")
    leia(sexo)
    escreva("Digite a altura: ")
    leia(altura)
    escreva("Digite o peso: ")
    leia(peso)
    se (sexo = "M") entao
        peso_ideal <- (72.7 * altura) - 58
    senao
        peso_ideal <- (62.1 * altura) - 44.7
    fimse
    escreval("Peso ideal = ", peso_ideal)
    se (peso > peso_ideal) entao
        escreval("ACIMA DO PESO IDEAL!")
    senao
        escreval("DENTRO DO PESO IDEAL!")
    fimse
```

fimalgoritmo

17-) Uma empresa concederá um aumento de salário aos seus funcionários, variável de acordo com o cargo, conforme a tabela abaixo. Faça um algoritmo que leia o salário e o cargo de um funcionário e calcule o novo salário. Se o cargo do funcionário não estiver na tabela, ele deverá, então, receber 40% de aumento. Mostre o salário antigo, o novo salário e o aumento concedido.

Código	Cargo	Percentual
101	Gerente	10%
102	Engenheiro	20%
103	Técnico	30%

```
algoritmo "Ex17"
var
salario_antigo, salario_novo, aumento: real
codigo: inteiro
inicio
    escreva("Digite o código do cargo: ")
    leia(codigo)
    escreva("Digite o salário atual: ")
    leia(salario_antigo)
    se (codigo = 101) entao
        aumento <- salario_antigo * 10/100
        salario_novo <- salario_antigo * aumento
    senao
        se (codigo = 102) entao
            aumento <- salario_antigo * 20/100
            salario_novo <- salario_antigo + aumento
        senao
            se (codigo = 103) entao
                aumento <- salario_antigo * 30/100
                salario_novo <- salario_antigo + aumento
            senao
                aumento <- salario_antigo * 40/100
                salario_novo <- salario_antigo + aumento
            fimse
        fimse
    fimse
    escreval("Salário antigo: ", salario_antigo)
    escreval("Novo salário: ", salario_novo)
    escreval("Aumento: ", aumento)
fimalgoritmo
```

18-) Escrever um algoritmo que leia um conjunto de 3 valores *ordem*, *a* e *b*, onde *ordem* é um valor inteiro e positivo e *a* e *b* são quaisquer valores reais. Escreva-os conforme definido a seguir:

- a) Se $ordem=1$ escrever os valores a e b em ordem crescente.
b) Se $ordem=2$ escrever os valores a e b em ordem decrescente.

```
algoritmo "Ex18"
var
a, b, n1, n2: real
ordem: inteiro
inicio
    escreva("Digite o 1o valor: ")
    leia(a)
    escreva("Digite o 2o valor: ")
    leia(b)
    escreva("Digite a ordem (1-cresc / 2-decresc): ")
    leia(ordem)
    se (a > b) entao
        n1 <- b
        n2 <- a
    senao
        n1 <- a
        n2 <- b
    fimse
    se (ordem = 1) entao
        escreval("Ordem crescente: ", n1, ", ", n2)
    senao
        escreval("Ordem decrescente: ", n2, ", ", n1)
    fimse
fimalgoritmo
```

19-) Escrever um algoritmo que leia o RA, as 2 notas obtidas por um aluno nas e a média das atividades (ATPS) que fazem parte da avaliação. Calcular a média de aproveitamento (MA), usando a fórmula:

$$MA = (Nota1 \times 4 + Nota2 \times 6) / 10 \times 70\% + ATPS \times 30\%$$

A atribuição de conceitos obedece à tabela abaixo:

Média de Aproveitamento	Conceito
$\geq 9,0$	A
$\geq 7,5$ e $< 9,0$	B
$\geq 6,0$ e $< 7,5$	C
$\geq 4,0$ e $< 6,0$	D
$< 4,0$	E

O algoritmo deve escrever o RA do aluno, suas notas, a média das atividades, a média de aproveitamento, o conceito correspondente e a mensagem: APROVADO se o conceito for A, B ou C e REPROVADO se for D ou E.

```
algoritmo "Ex19"
var
ra, conceito: caractere
nota1, nota2, ma, atps: real
inicio
    escreva("Digite o RA:")
    leia(ra)
    escreva("Digite a nota 1:")
    leia(nota1)
    escreva("Digite a nota 2:")
    leia(nota2)
    escreva("Digite a nota das atividades:")
    leia(atps)
    ma <- (nota1*4 + nota2*6)/10*70/100 + (atps*30/100)
    escreval("RA Aluno: ", ra)
    escreval("Nota 1: ", nota1)
    escreval("Nota 2: ", nota2)
    escreval("Média Atividades: ", atps)
    escreval("Média Aproveitamento: ", ma)
    se (ma >= 9.0) entao
        conceito <- "A"
    senao
        se (ma >= 7.5) entao
            conceito <- "B"
        senao
            se (ma >= 6.0) entao
                conceito <- "C"
            senao
                se (ma >= 4.0) entao
                    conceito <- "D"
                senao
                    conceito <- "E"
                fimse
            fimse
        fimse
    fimse
    escreval("Conceito: ", conceito)
    se (conceito = "D") entao
        escreval("REPROVADO")
    senao
        se (conceito = "E") entao
            escreval("REPROVADO")
        senao
            escreval("APROVADO")
        fimse
    fimse
fimalgoritmo
```

20-) O departamento que controla o índice de poluição do meio ambiente mantém 3 grupos de indústrias que são altamente poluentes. O índice de poluição aceitável varia de 0,05 até 0,25. Se o índice sobe para 0,3 as indústrias do 1º grupo são intimadas a suspenderem suas atividades, se o índice cresce para 0,4 as do 1º e 2º grupo são intimadas a suspenderem suas atividades e se o índice atingir 0,5 todos os 3 grupos devem ser notificados a paralisarem suas atividades. Escrever um algoritmo que leia o índice de poluição medido e emita a notificação adequada aos diferentes grupos de empresas.

```
algoritmo "Ex20"
var
indice: real
inicio
    escreva("Digite o Índice de Poluição: ")
    leia(indice)
    se (indice >= 0.3) entao
        escreval("Empr. do Grupo 1 - PARALISAR OPERAÇÕES")
    fimse
    se (indice >= 0.4) entao
        escreval("Empr. do Grupo 2 - PARALISAR OPERAÇÕES")
    fimse
    se (indice >= 0.5) entao
        escreval("Empr. do Grupo 3 - PARALISAR OPERAÇÕES")
    fimse
fimalgoritmo
```