

ATIVIDADE EXTRA-CLASSE**6 – Estrutura de Dados: Vetor, Matriz e Registro**
GABARITO

A-) Escreva algoritmos para solução dos problemas descritos.

1. Faça um algoritmo que leia 10 números inteiros e armazene-os em um vetor, depois copie estes valores de um vetor para um segundo vetor e depois, mostre-os impresso.

```
algoritmo "EX01"
var
    vet1: vetor[1..10] de inteiro
    vet2: vetor[1..10] de inteiro
    i: inteiro
inicio
    para i de 1 ate 10 passo 1 faca
        escreva("Digite o valor #", i, ": ")
        leia(vet1[i])
        vet2[i] <- vet1[i]
    fimpara
    para i de 1 ate 10 passo 1 faca
        escreval("Vetor2[" , i, "] = " , vet2[i])
    fimpara
fimalgoritmo
```

2. Elabore um algoritmo que leia a média de 10 alunos, armazenando estas médias em um vetor e depois de lido calcule a média geral, e mostre o valor calculado ao final.

```
algoritmo "EX02"
var
    media: vetor[1..10] de real
    mediageral: real
    i: inteiro
inicio
    mediageral <- 0
    para i de 1 ate 10 passo 1 faca
        escreva("Digite a média #", i, ": ")
        leia(media[i])
    fimpara
    para i de 1 ate 10 passo 1 faca
        mediageral <- mediageral + media[i]
    fimpara
    mediageral <- mediageral / 10
    escreval("Média geral = " , mediageral)
fimalgoritmo
```

3. Altere o algoritmo anterior mostrando também o número de alunos abaixo da média e o número de alunos maior ou igual à média calculada.

```
algoritmo "EX03"
var
    media: vetor[1..10] de real
    mediageral: real
    i, mediaacima, mediaabaixo: inteiro
inicio
    mediageral <- 0
    mediaacima <- 0
    mediaabaixo <- 0
    para i de 1 ate 10 passo 1 faca
        escreva("Digite a média #", i, ": ")
        leia(media[i])
    fimpara
    para i de 1 ate 10 passo 1 faca
        mediageral <- mediageral + media[i]
    fimpara
    mediageral <- mediageral / 10
    para i de 1 ate 10 passo 1 faca
        se (media[i] >= mediageral) entao
            mediaacima <- mediaacima + 1
        senao
            mediaabaixo <- mediaabaixo + 1
        fimse
    fimpara
    escreval("Média geral = ", mediageral)
    escreval("Médias acima = ", mediaacima)
    escreval("Médias abaixo = ", mediaabaixo)
fimalgoritmo
```

4. Faça um algoritmo que some o conteúdo de dois vetores e armazene o resultado em um terceiro vetor.

```
algoritmo "EX04"
var
    vet1: vetor[1..10] de inteiro
    vet2: vetor[1..10] de inteiro
    vet3: vetor[1..10] de inteiro
    i: inteiro
inicio
    para i de 1 ate 10 passo 1 faca
        escreva("Digite o valor do Vetor 1(", i, "): ")
        leia(vet1[i])
        escreva("Digite o valor do Vetor 2(", i, "): ")
        leia(vet2[i])
    fimpara
    para i de 1 ate 10 passo 1 faca
        vet3[i] <- vet1[i] + vet2[i]
```

```
    escreval("Vetor 3[", i, "] = ", vet3[i])
fimpara
fimalgoritmo
```


5. Elabore um algoritmo que leia a venda dos carros de uma concessionária durante uma semana padrão (2ª a Sábado), armazenando num vetor, depois mostre:

- a média de vendas,
- o dia da semana com a maior venda de veículos registrado;
- o dia da semana com a menor venda de veículos registrado.

```
algoritmo "EX05"
var
    venda: vetor[1..10] de inteiro
    media: real
    maior, menor, i: inteiro
inicio
    media <- 0;
    para i de 1 ate 6 passo 1 faca
        escreva("Digite o total de vendas #", i, ": ")
        leia(venda[i])
    fimpara
    para i de 1 ate 6 passo 1 faca
        media <- media + venda[i]
        se (i = 1) entao
            maior <- venda[i]
            menor <- venda[i]
        senao
            se (maior < venda[i]) entao
                maior <- venda[i]
            fimse
            se (menor > venda[i]) entao
                menor <- venda[i]
            fimse
        fimse
    fimpara
    media <- media / 6
    escreval("Média Geral = ", media)
    escreval("Maior Venda = ", maior)
    escreval("Menor Venda = ", menor)
fimalgoritmo
```

6. Faça um algoritmo para somar duas matrizes de tamanho 4 x 3.

A	B	A + B
_____	_____	_____



```

algoritmo "EX06"
var
    mat1: vetor[1..4,1..3] de inteiro
    mat2: vetor[1..4,1..3] de inteiro
    mat3: vetor[1..4,1..3] de inteiro
    i, j: inteiro
inicio
    para i de 1 ate 4 passo 1 faca
        para j de 1 ate 3 passo 1 faca
            escreva("Digite o item da Matriz 1 [",i,"",",",j,""]:"
")
            leia(mat1[i,j])
            escreva("Digite o item da Matriz 2 [",i,"",",",j,""]:"
")
            leia(mat2[i,j])
        fimpara
    fimpara
    para i de 1 ate 4 passo 1 faca
        para j de 1 ate 3 passo 1 faca
            mat3[i,j] <- mat1[i,j] + mat2[i,j]
            escreval("Matriz 3 [",i,"",",",j,""] = ", mat3[i,j])
        fimpara
    fimpara
fimalgoritmo
    
```

7. Faça um algoritmo que declare uma matriz com as notas/média de uma sala de 30 alunos, conforme definido a seguir:

		1	2	3	4	5	6	7	...	30
Nota 1	1									
Nota 2	2									
Média	3									

- Leia as 2 notas, preenchendo as linhas relativas às notas.
- Após ler todas as notas, faça o cálculo das médias dos alunos, conforme média padrão AEJ.
- Imprima o número do aluno, as 2 notas e sua média final.

```
algoritmo "EX07"
var
    mat: vetor[1..3,1..30] de real
    i: inteiro
inicio
    para i de 1 ate 10 passo 1 faca
        escreva("Nota 1 #", i, ": ")
        leia(mat[1, i])
        escreva("Nota 2 #", i, ": ")
        leia(mat[2, i])
    fimpara
    para i de 1 ate 10 passo 1 faca
        mat[3, i] <- mat[1, i] * 0.4 + mat[2, i] * 0.6
        escreval("Nota 1 do Aluno #", i, " = ", mat[1, i])
        escreval("Nota 2 do Aluno #", i, " = ", mat[2, i])
        escreval("Média do Aluno #", i, " = ", mat[3, i])
        escreval("-----")
    fimpara
fimalgoritmo
```

8. Amplie o exercício 5 anterior, de forma que tenhamos o registro de 4 tipos de veículos numa matriz, realizando as mesmas operações definidas.

```
algoritmo "EX08"
var
    venda: vetor[1..4,1..6] de inteiro
    media: real
    maior, menor, i, j: inteiro
inicio
    media <- 0;
    para i de 1 ate 4 passo 1 faca
        escreval("Veículo #", i)
        escreval("-----")
        para j de 1 ate 6 passo 1 faca
            escreva("Digite o total de vendas #", j, ": ")
            leia(venda[i,j])
        fimpara
    fimpara
    para i de 1 ate 4 passo 1 faca
        para j de 1 ate 6 passo 1 faca
            media <- media + venda[i,j]
            se ((i = 1) e (j = 1)) entao
                maior <- venda[i,j]
                menor <- venda[i,j]
            senao
                se (maior < venda[i,j]) entao
                    maior <- venda[i,j]
        fimse
    fimse
```

```

        se (menor > venda[i,j]) entao
            menor <- venda[i,j]
        fimse
    fimse
fimpara
fimpara
media <- media / 24
escreval("-----")
escreval("Média Geral = ", media)
escreval("Maior Venda = ", maior)
escreval("Menor Venda = ", menor)
fimalgoritmo

```

9. Faça um algoritmo que leia 5 jogos da Mega-Sena em uma matriz e o gabarito (resultado de um jogo) da Mega-Sena em um vetor, conforme abaixo.

Resultado						
Jogo 1						
Jogo 2						
Jogo 3						
Jogo 4						
Jogo 5						

Após lidos os valores, faça a pontuação de cada jogo mostrando o resultado conforme abaixo:

- 0 – Nenhum acerto!
- 1 – 1 Acerto apenas!
- 2 – Duque!
- 3 – Terno!
- 4 – Quadra, retire seu prêmio!
- 5 – Quina, quase, retire seu prêmio!
- 6 – Sena, parabéns novo MILIONÁRIO!

```

algoritmo "EX09"
var
    i, j, num_acertos: inteiro
    msena: vetor[1..6] de inteiro
    jogos: vetor[1..5,1..6] de inteiro
inicio
    para i de 1 ate 5 passo 1 faca
        escreval("Jogo #", i)

```

```
    escreval("-----")
    para j de 1 ate 6 passo 1 faca
        escreva("Digite o ", j, "º: ")
        leia(jogos[i,j])
    fimpara
fimpara
escreval("Mega-Sena")
escreval("-----")
para i de 1 ate 6 passo 1 faca
    escreva("Digite o Número ", i, " Sorteado : ")
    leia(msena[i])
fimpara
escreval("-----")
escreval("RESULTADOS: ")
escreval("-----")
para i de 1 ate 5 passo 1 faca
    num_acertos <- 0
    escreval("Jogo #", i)
    escreval("-----")
    para j de 1 ate 6 passo 1 faca
        se (jogos[i,j] = msena[j]) entao
            num_acertos <- num_acertos + 1
        fimse
    fimpara
    escolha (num_acertos)
        caso 0
            escreval("Nenhum acerto!")
        caso 1
            escreval("1 Acerto apenas!")
        caso 2
            escreval("Duque!")
        caso 3
            escreval("Terno!")
        caso 4
            escreval("Quadra, retire seu prêmio!")
        caso 5
            escreval("Quina, quase, retire seu prêmio!")
        outrocaso
            escreval("Sena, parabéns novo MILIONÁRIO!")
    fimescolha
    escreval("-----")
fimpara
fimalgoritmo
```

10. Fazer um algoritmo que leia 10 valores em uma matriz e, depois de lido, ordene os valores e imprima as duas listagens ao final: lista de números não-ordenados e lista de números ordenados.

	1	2	3	4	5	6	7	8	9	10
Não-ordenada										
Ordenada										

```

algoritmo "EX10"
var
  i, j, tempo: inteiro
  mat: vetor[1..2,1..10] de inteiro
inicio
  para i de 1 ate 10 passo 1 faca
    escreva("Digite o número ", i, ": ")
    leia(mat[1,i])
    mat[2,i] <- mat[1,i]
  fimpara
  para i de 1 ate 10 passo 1 faca
    para j de i+1 ate 10 passo 1 faca
      se (mat[2,i] > mat[2,j]) entao
        tempo <- mat[2,i]
        mat[2,i] <- mat[2,j]
        mat[2,j] <- tempo
      fimse
    fimpara
  fimpara
  escreval("Números Não-Ordenados: ")
  escreval("-----")
  para i de 1 ate 10 passo 1 faca
    escreva(mat[1,i], " ")
  fimpara
  escreval()
  escreval("Números Ordenados: ")
  escreval("-----")
  para i de 1 ate 10 passo 1 faca
    escreva(mat[2,i], " ")
  fimpara
  escreval()
fimalgoritmo

```

11. Escreva um programa para cadastrar 1 aluno. As informações necessárias são: *nome do aluno*, *nome da disciplina* e *média final*. Deve ser usada uma estrutura heterogênea para a construção deste cadastro, ou seja, declaração do registro. Após o cadastro, o registro do aluno com valores deve ser impresso na tela.

```

algoritmo "EX11"
var
  tipo ALUNO = registro
  nome, disciplina: caractere

```



```
mediafinal: real
fim registro
  pupilo: ALUNO
inicio
  escreva("Nome aluno: ")
  leia(pupilo.nome)
  escreva("Disciplina: ")
  leia(pupilo.disciplina)
  escreva("Média: ")
  leia(pupilo.mediafinal)
  escreva("-----")
  escreva("Nome aluno: ", pupilo.nome)
  escreva("Disciplina: ", pupilo.disciplina)
  escreva("Média: ", pupilo.mediafinal)
finalgoritmo
```

12. Reescreva o programa anterior para cadastrar até 30 alunos de uma turma. Ao final do cadastro de cada aluno deverá ser perguntado: "Novo Aluno (S/N)?".

Deve-se utilizar um vetor do tipo declarado como registro para a solução deste programa. Após o término de todos os cadastros, ou seja, quando o usuário digitar "N" na pergunta final, deverá ser perguntado: "Exibir Aluno [1-30]?", e após digitado o número correspondente, a ficha do mesmo será impressa na tela.

```
algoritmo "EX12"
var
tipo ALUNO = registro
  nome, disciplina: caractere
  mediafinal: real
fim registro
  pupilo: vetor[1..30] de ALUNO
  cont: inteiro
  opcao: caractere
inicio
  cont <- 0
  repita
    escreva("Novo Aluno (S/N)? ")
    leia(opcao)
    se (opcao = "S") entao
      cont <- cont + 1
      escreva("-----")
      escreva("Aluno #", cont)
      escreva("-----")
      escreva("Nome aluno: ")
      leia(pupilo[cont].nome)
      escreva("Disciplina: ")
      leia(pupilo[cont].disciplina)
```

```
        escreva("Média: ")
        leia(pupilo[cont].mediafinal)
        escreva("-----")
    fimse
ate ((cont = 30) ou (opcao = "N"))
repita
    escreva("Exibir Aluno (1-30)? ")
    leia(cont)
    se (cont <= 30) entao
        escreva("-----")
        escreva("Aluno #", cont)
        escreva("-----")
        escreva("Nome aluno: ", pupilo[cont].nome)
        escreva("Disciplina: ", pupilo[cont].disciplina)
        escreva("Média: ", pupilo[cont].mediafinal)
        escreva("-----")
    fimse
ate (cont > 30)
fimalgoritmo
```