

**ATIVIDADE EXTRA-CLASSE****5 – Estrutura de Iteração  
GABARITO**

Escreva algoritmos para solução dos problemas descritos utilizando a ferramenta VisualG. Utilize as seguintes estruturas:

- ⇒ Exercícios de 1 a 7 com ENQUANTO..FACA
- ⇒ Exercícios de 8 a 14 com REPITA..ATE
- ⇒ Exercícios de 15 a 20 com PARA..FACA

1. Imprimir na tela os números de 0 a 100.

```
algoritmo "EX01"
var
    num: inteiro
inicio
    num <- 0
    enquanto (num <= 100) faca
        escreval(num)
        num <- num + 1
    fimenquanto
fimalgoritmo
```

2. Imprimir na tela os números de 0 a 100, com o rótulo a seguir:

>> 0, 1, 2, 3, 4, 5, 6, ..., 97, 98, 99, 100 <<

```
algoritmo "EX02"
var
    num: inteiro
inicio
    num <- 0
    escreva(">> ")
    enquanto (num <= 99) faca
        escreva(num, ", ")
        num <- num + 1
    fimenquanto
    escreva (num, " <<")
fimalgoritmo
```

3. Ler 20 valores do teclado e imprimir na tela estes valores, conforme rótulo a seguir:

**Valor #1: x**

Valor #2: y  
Valor #3: z  
...  
Valor #20: w

```
algoritmo "EX03"
var
    num, valor: inteiro
inicio
    num <- 0
    enquanto (num < 20) faca
        escreva("Digite um valor: ")
        leia(valor)
        num <- num + 1
        escreval("Valor #", num, ": ", valor)
    fimenquanto
fimalgoritmo
```

4. Ler 20 valores do teclado e imprimir na tela estes valores, e ao final a soma dos seus valores, conforme rótulo a seguir:

Valor #1: x  
Valor #2: y  
Valor #3: z  
...  
Valor #20: w

-----  
**SOMA = xxxxx**

```
algoritmo "EX04"
var
    num, valor, soma: inteiro
inicio
    num <- 0
    soma <- 0
    enquanto (num < 20) faca
        escreva("Digite um valor: ")
        leia(valor)
        soma <- soma + valor
        num <- num + 1
        escreval("Valor #", num, ": ", valor)
    fimenquanto
    escreval("-----")
    escreval("SOMA = ", soma)
fimalgoritmo
```

5. Ler 20 valores do teclado e imprimir na tela se este número é par ou ímpar e ao final o total de números pares e o total de números ímpares.

```
algoritmo "EX05"
var
    num, valor, totalPar, totalImpar: inteiro
inicio
    num <- 0
    totalPar <- 0
    totalImpar <- 0
    enquanto (num < 20) faca
        escreva("Digite um valor: ")
        leia(valor)
        se (valor MOD 2 = 0) entao
            escreval("Número Par!")
            totalPar <- totalPar + 1
        senao
            escreval("Número Ímpar!")
            totalImpar <- totalImpar + 1
        fimse
        num <- num + 1
    fimenquanto
    escreval("-----")
    escreval("PARES = ", totalPar)
    escreval("ÍMPARES = ", totalImpar)
finalgoritmo
```

6. Ler 20 valores do teclado e imprimir na tela se o número é par ou ímpar e também positivo ou negativo e ao final os rótulos de totais a seguir:

**Total de Números Pares Positivos: pp**  
**Total de Números Pares Negativos: pn**  
**Total de Números Ímpares Positivos: ip**  
**Total de números Ímpares Negativos: in**

```
algoritmo "EX06"
var
    num, valor, pp, pn, ip, in: inteiro
inicio
    num <- 0
    pp <- 0
    pn <- 0
    ip <- 0
    in <- 0
    enquanto (num < 20) faca
        escreva("Digite um valor: ")
        leia(valor)
        se (valor MOD 2 = 0) entao
```

```
se (valor >= 0) entao
    pp <- pp + 1;
    escreval("Número Par e Positivo!")
senao
    escreval("Número Par e Negativo!")
    pn <- pn + 1
fimse
senao
    se (valor >= 0) entao
        ip <- ip + 1
        escreval("Número Ímpar e Positivo!")
    senao
        escreval("Número Ímpar e Negativo!")
        in <- in + 1
    fimse
fimse
num <- num + 1
fimenquanto
escreval("-----")
escreval("Total de Números Pares Positivos: ", pp)
escreval("Total de Números Pares Negativos: ", pn)
escreval("Total de Números Ímpares Positivos: ", ip)
escreval("Total de números Ímpares Negativos: ", in)
finalgoritmo
```

7. Faça um algoritmo que receba a idade e sexo de "N" pessoas e mostre mensagem informando “*maior de idade*” e “*menor de idade*” para cada pessoa (considere a idade a partir de 18 anos como maior de idade); mostre também mensagem informando se a pessoa é "Homem" ou "Mulher". Mostre ao final também:

**Núm. Pessoas Maiores de Idade: xx**  
**Núm. Pessoas Menores de Idade: yy**  
**Núm. de Homens: th**  
**Núm. de Mulheres: tm**  
**Média das Idades dos Homens: mh**  
**Média das Idades das Mulheres: mm**  
**Média Total das Idades: mt**

```
algoritmo "EX07"
var
    num, numPessoas, idade, xx, yy, th, tm, sh, sm: inteiro
    sexo: caractere
    mh, mm, mt: real
inicio
    num <- 0
    xx <- 0
    yy <- 0
```

```
th <- 0
tm <- 0
sh <- 0
sm <- 0
escreva("Quantas pessoas deseja ler: ")
leia(numPessoas)
enquanto (num < numPessoas) faça
    escreva("Digite a idade: ")
    leia(idade)
    escreva("Digite o sexo: ")
    leia(sexo)
    se (idade >= 18) entao
        escreval("Maior de idade!")
        xx <- xx + 1
    senao
        escreval("Menor de idade!")
        yy <- yy + 1
    fimse

    se (sexo = "M") entao
        escreval("Homem!")
        sh <- sh + idade
        th <- th + 1
    senao
        escreval("Mulher!")
        sm <- sm + idade
        tm <- tm + 1
    fimse
    num <- num + 1
fimenquanto
mh <- sh / th
mm <- sm / tm
mt <- (sh + sm) / (th + tm)
escreval("-----")
escreval("Núm. Pessoas Maiores de Idade: ", xx)
escreval("Núm. Pessoas Menores de Idade: ", yy)
escreval("Núm. de Homens: ", th)
escreval("Núm. de Mulheres: ", tm)
escreval("Média das Idades dos Homens: ", mh)
escreval("Média das Idades das Mulheres: ", mm)
escreval("Média Total das Idades: ", mt)
fimalgoritmo
```

8. Escreva um algoritmo que leia o código de um aluno e suas 2 notas. Calcule a média ponderada do aluno (padrão AEJ). Mostre o código do aluno, suas notas, a média calculada e uma mensagem "APROVADO" se a média for maior ou igual a 5 e "REPROVADO" se a média for menor que 5. Repita a operação até que o código lido seja negativo.

```
algoritmo "EX08"
var
  ra: inteiro
  nota1, nota2, media: real
inicio
  repita
    escreva("Digite seu RA: ")
    leia(ra)
    escreva("Digite a nota 1: ")
    leia(nota1)
    escreva("Digite a nota 2: ")
    leia(nota2)
    media <- ((nota1 * 4) + (nota2 * 6)) / 10
    se (media >= 5) entao
      escreval("APROVADO")
    senao
      escreval("REPROVADO")
    fimse
  ate (ra < 0)
finalgoritmo
```

9. Foi feita uma pesquisa entre os 1000 habitantes de uma região para coletar os seguintes dados: sexo (0-feminino, 1-masculino), idade e altura. Faça um algoritmo que leia as informações coletadas e mostre as seguintes informações:

- média da idade do grupo;
- média da altura das mulheres;
- média da idade dos homens;
- percentual de pessoas com idade entre 18 e 35 anos (inclusive).

```
algoritmo "EX09"
var
  cont, sexo, idade, ti, tih, tif, tm, th: inteiro
  altura, mi, mam, tam, mih, pif: real
inicio
  cont <- 1
  th <- 0
  tm <- 0
  ti <- 0
  tih <- 0
  tif <- 0
  tam <- 0
  repita
    escreval("Pesquisa #", cont)
    escreva("Digite seu sexo (0-M / 1-F): ")
    leia(sexo)
    escreva("Digite a idade: ")
```

```

leia(idade)
escreva("Digite a altura: ")
leia(altura)
ti <- ti + idade
se ((idade >= 18) e (idade <= 35)) entao
    tif <- tif + 1
fimse
se (sexo = 0) entao
    th <- th + 1
    tih <- tih + idade
senao
    tm <- tm + 1
    tam <- tam + altura
fimse
cont <- cont + 1
ate (cont > 1000)
mi <- ti / 1000
mam <- tam / tm
mih <- tih / th
pif <- (tif / 1000) * 100
escreval("-----")
escreval("Média da idade do grupo = ", mi)
escreval("Média da altura das mulheres = ", mam)
escreval("Média da idade dos homens = ", mih)
escreval("Percentual de pessoas com idade entre 18 e 35
anos = ", pif, "%")
fimalgoritmo

```

10. Escrever um algoritmo que leia um conjunto de 50 informações contendo, cada uma delas, a altura e o sexo de uma pessoa (código=1, masculino código=2, feminino), calcule e mostre o seguinte:

- a maior e a menor altura da turma
- a média da altura das mulheres
- a média da altura da turma

```

algoritmo "EX10"
var
    cont, sexo, tfem: inteiro
    altura, maalt, mealt, mtur, mfem, stur, sfem: real
inicio
    cont <- 1
    tfem <- 0
    stur <- 0
    sfem <- 0
    repita
        escreval("Informação #", cont)
        escreva("Digite seu sexo (1-M / 2-F): ")

```

```

leia(sexo)
escreva("Digite a altura: ")
leia(altura)
stur <- stur + altura
se (cont = 1) entao
    maalt <- altura
    mealt <- altura
senao
    se (altura > maalt) entao
        maalt <- altura
    fimse
    se (altura < mealt) entao
        mealt <- altura
    fimse
fimse
se (sexo = 2) entao
    sfem <- sfem + altura
    tfem <- tfem + 1
fimse
cont <- cont + 1
ate (cont > 50)
mfem <- sfem / tfem
mtur <- stur / 50
escreval("-----")
escreval("Maior altura do grupo = ", maalt)
escreval("Menor altura do grupo = ", mealt)
escreval("Média de altura mulheres = ", mfem)
escreval("Média de altura do grupo = ", mtur)
fimalgoritmo

```

11. A concessionária de veículos “CARANGO VELHO” está vendendo os seus veículos com desconto. Faça um algoritmo que leia o valor de um veículo e calcule e exiba o valor do desconto e o valor a ser pago pelo cliente (*valor do veículo – desconto*) de vários carros. O desconto deverá ser calculado de acordo com o ano do veículo, ou seja:

- Até 2000 - 12% e
- Acima de 2000 - 7%.

O sistema deverá perguntar se deseja continuar calculando desconto até que a resposta seja: “(N) Não”. Informar total de carros com ano até 2000 vendidos.

```

algoritmo "EX11"
var
    ano, t2000: inteiro
    vlcarro, pdesc, vl-desc, vltotal: real
    opcao: caractere
inicio
    t2000 <- 0

```



```

escreval("CARANGO VELHO")
repita
    escreva("Digite o Ano do Veículo: ")
    leia(ano)
    escreva("Digite o Valor do Veículo: ")
    leia(vlcarro)
    se (ano > 2000) entao
        pdesc <- 7
    senao
        pdesc <- 12
        t2000 <- t2000 + 1
    fimse
    vldesc <- vlcarro * pdesc / 100
    vltotal <- vlcarro - vldesc
    escreval("-----")
    escreval("Desconto: R$ ", vldesc)
    escreval("Valor Total: R$ ", vltotal)
    escreval("-----")
    escreva("Vender novo veículo (S/N)? ")
    leia(opcao)
    ate (opcao = "N")
    escreval("-----")
    escreval("Total de carros até 2000 vendidos = ", t2000)
finalgoritmo

```

12. O Ministério da Defesa brasileiro montou o seguinte sistema que será divulgado em todas as juntas militares do Brasil:

### Serviço Militar

```

<1>..Aeronáutica
<2>..Exército
<3>..Marinha
<0>..Finalizar

```

```

=====
Opção (0-3): ____
=====

```

Após o funcionário escolher a opção desejada (1-3), o sistema deverá:

- Perguntar a idade da pessoa e validá-la, ou seja, se o candidato estiver entre 17 e 19 anos, considerar o candidato, caso contrário, mostrar mensagem: "Idade Inválida!" e não considerar este candidato.
- Perguntar o sexo.
- Mostrar novamente o menu.

Quando a opção digitada for 0, o sistema será finalizado e mostrará:

- Número total de pessoas registradas e número total de pessoas inválidas.
- Número de Homens e Mulheres registradas em cada área: Aeronáutica, Exército e Marinha.

```
algoritmo "EX12"
var
    idade, nv, ni, ha, he, hm, ma, me, mm, opcao: inteiro
    sexo: caractere
inicio
    nv <- 0
    ni <- 0
    ha <- 0
    he <- 0
    hm <- 0
    ma <- 0
    me <- 0
    mm <- 0
    repita
        escreval("Serviço Militar")
        escreval("-----")
        escreval("<1>..Aeronáutica")
        escreval("<2>..Exército")
        escreval("<3>..Marinha")
        escreval("<0>..Finalizar")
        escreval("=====")
        escreva("Opção (0-3): ")
        leia(opcao)
        escreval("=====")
        se (opcao > 0) entao
            escreva("Digite a Idade: ")
            leia(idade)
            se ((idade >= 17) e (idade <= 19)) entao
                nv <- nv + 1
                escreva("Digite o sexo (M/F): ")
                leia(sexo)
                se (sexo = "M") entao
                    escolha (opcao)
                    caso 1
                        ha <- ha + 1
                    caso 2
                        he <- he + 1
                    outrocaso
                        hm <- hm + 1
                    fimescolha
                senao
                    escolha (opcao)
                    caso 1
                        ma <- ma + 1
```

```

        caso 2
            me <- me + 1
        outrocaso
            mm <- mm + 1
        fimsecolha
    fimse
senao
    ni <- ni + 1
    escreval("Idade Inválida!")
fimse
fimse
ate (opcao = 0)
escreval("-----")
escreval("Pessoas Registradas = ", nv)
escreval("Pessoas Inválidas = ", ni)
escreval("Homens na Aeronáutica = ", ha)
escreval("Homens no Exército = ", he)
escreval("Homens na Marinha = ", hm)
escreval("Mulheres na Aeronáutica = ", ma)
escreval("Mulheres no Exército = ", me)
escreval("Mulheres na Marinha = ", mm)
fimalgoritmo

```

13. Foi feita uma pesquisa entre os habitantes de uma região. Foram coletados os dados de idade, sexo (M/F) e salário. Faça um algoritmo que informe:

- a média de salário do grupo;
- maior e menor idade do grupo;
- quantidade de mulheres com salário até R\$678,00.

Encerre a entrada de dados quando for digitada uma idade negativa.

```

algoritmo "EX13"
var
    cont, idade, maiorid, menorid, numsal: inteiro
    sexo: caractere
    msal, ssal, salario: real
inicio
    cont <- 0
    ssal <- 0
    numsal <- 0
    repita
        escreval("-----")
        escreva("Digite sua idade: ")
        leia(idade)
        se (idade >= 0) entao
            escreva("Digite seu sexo: ")
            leia(sexo)
            escreva("Digite seu salário: ")

```

```
leia(salario)
cont <- cont + 1
ssal <- ssal + salario
se (cont = 1) entao
    maiorid <- idade
    menorid <- idade
senao
    se (maiorid < idade) entao
        maiorid <- idade
    fimse
    se (menorid > idade) entao
        menorid <- idade
    fimse
fimse
se ((sexo = "F") e (salario <= 678)) entao
    numsal <- numsal + 1
fimse
fimse
ate (idade < 0)
escreval("-----")
msal <- ssal / cont
escreval("Média de salário do grupo = ", msal)
escreval("Maior idade do grupo = ", maiorid)
escreval("Menor idade do grupo = ", menorid)
escreval("Qtd. mulheres com salário até 678 = ", numsal)
fimalgoritmo
```

14. Um restaurante montou um cardápio online, com um terminal na mesa de cada cliente. Faça um algoritmo que mostre o cardápio a seguir:

#### Comidas

```
-----
<1>...Strogonoff com Palmito - 49,95
<2>...Milanesa de Funghi - 35,79
<3>...Pizza de Alho - 37,89
<4>...Bruschetta a la Moda - 35,17
<5>...Salada Milano - 17,95
<6>...Salada Caesar - 19,05
```

#### Bebidas

```
-----
<7>...Vinho da Casa (Jarra) - 29,50
<8>...Refrigerante - 4,50
<9>...Água - 2,50
<10>..Sucos Naturais - 6,50
<11>..Cerveja - 7,00
<12>..Chopp (500 ml) - 8,50
```

#### Sobremesa

```
-----
```

```

<13>..Tiramisú - 12,00
<14>..Petit Gateau - 15,00
<15>..Papaia com Cassis - 16,50

```

```

=====
Escolha Opção (1-15): ____
=====

```

Após a pessoa escolher a opção, deve-se perguntar a quantidade desejada e mostrar novamente o cardápio; caso a pessoa escolha um valor diferente das opções, o sistema deverá finalizar e imprimir o total do pedido.

```

algoritmo "EX14"
var
    qtd, opcao: inteiro
    total, valor: real
inicio
    total <- 0
    repita
        escreval("-----")
        escreval("Comidas")
        escreval("-----")
        escreval("<1>...Strogonoff com Palmito - 49,95")
        escreval("<2>...Milanesa de Funghi - 35,79")
        escreval("<3>...Pizza de Alho - 37,89")
        escreval("<4>...Bruschetta a la Moda - 35,17")
        escreval("<5>...Salada Milano - 17,95")
        escreval("<6>...Salada Caesar - 19,05")
        escreval("-----")
        escreval("Bebidas")
        escreval("-----")
        escreval("<7>...Vinho da Casa (Jarra) - 29,50")
        escreval("<8>...Refrigerante - 4,50")
        escreval("<9>...Água - 2,50")
        escreval("<10>..Sucos Naturais - 6,50")
        escreval("<11>..Cerveja - 7,00")
        escreval("<12>..Chopp (500 ml) - 8,50")
        escreval("-----")
        escreval("Sobremesa")
        escreval("-----")
        escreval("<13>..Tiramisú - 12,00")
        escreval("<14>..Petit Gateau - 15,00")
        escreval("<15>..Papaia com Cassis - 16,50")
        escreval("=====")
        escreva("Escolha Opção (1-15): ")
        leia(opcao)
        escreval("=====")
        se ((opcao >= 1) e (opcao <= 15)) entao
            escreva("Digite a Quantidade: ")
            leia(qtd)

```

```
escolha (opcao)
  caso 1
    valor <- 49.95
  caso 2
    valor <- 35.79
  caso 3
    valor <- 37.89
  caso 4
    valor <- 35.17
  caso 5
    valor <- 17.95
  caso 6
    valor <- 19.05
  caso 7
    valor <- 29.50
  caso 8
    valor <- 4.50
  caso 9
    valor <- 2.50
  caso 10
    valor <- 6.50
  caso 11
    valor <- 7.00
  caso 12
    valor <- 8.50
  caso 13
    valor <- 12.00
  caso 14
    valor <- 15.00
  outrocaso
    valor <- 16.50
fimescolha
total <- total + (qtd * valor)
fimse
ate ((opcao < 1) ou (opcao > 15))
escreval("Total do Pedido = ", total)
escreval("=====")
fimalgoritmo
```

15. A progressão aritmética é dado por um termo inicial  $a_1$ , um termo final  $a_f$  e o valor da  $PA$ . Pegando como exemplo um termo inicial  $a_1 = 5$ , um termo final  $a_f = 17$  e a  $PA = 4$ , teríamos  $PA = (5, 9, 13, 17)$ . Monte um algoritmo que leia os termos iniciais  $a_1$ ,  $a_f$  e  $PA$  e imprima na tela a sequência desta progressão aritmética.

```
algoritmo "EX15"
var
  a1, af, pa, i: inteiro
```

```
inicio
  escreva("Digite o termo inicial a1: ")
  leia(a1)
  escreva("Digite o termo final af: ")
  leia(af)
  escreva("Digite o valor da PA: ")
  leia(pa)
  para i de a1 ate af passo pa faca
    escreval(i)
  fimpara;
fimalgoritmo
```

16. Faça um algoritmo que receba o preço de custo e o preço de venda de 10 produtos. Mostre como resultado se houve lucro, prejuízo ou empate para cada produto. Informe media de preço de custo e do preço de venda.

```
algoritmo "EX16"
var
  i: inteiro
  custo, venda, lucro: real
inicio
  para i de 1 ate 10 passo 1 faca
    escreval("Produto #", i)
    escreval("-----")
    escreva("Digite o preço de custo: ")
    leia(custo)
    escreva("Digite o preço de venda: ")
    leia(venda)
    lucro <- venda - custo
    se (lucro > 0) entao
      escreval("Lucro de ", lucro)
    senao
      se (lucro < 0) entao
        escreval("Prejuízo de ", lucro)
      senao
        escreval("Não houve lucro nem prejuízo!")
      fimse
    fimse
  fimpara
fimalgoritmo
```

17. Faça um programa que receba as notas (n1,n2) de 10 alunos, calcule e mostre a média de cada aluno. Ao final mostre a média geral de todos os 10 alunos, a pior e a melhor média.

```
algoritmo "EX17"
var
```

```

i: inteiro
n1, n2, media, piormedia, melhormedia, mediageral: real
inicio
  mediageral <- 0
  para i de 1 ate 10 passo 1 faca
    escreval("Aluno #", i)
    escreval("-----")
    escreva("Digite a nota 1: ")
    leia(n1)
    escreva("Digite a nota 2: ")
    leia(n2)
    media <- (n1 * 4 + n2 * 6) / 10
    escreval("-----")
    escreval("Média = ", media)
    escreval("-----")
    mediageral <- mediageral + media
    se (i = 1) entao
      piormedia <- media
      melhormedia <- media
    senao
      se (piormedia > media) entao
        piormedia <- media
      fimse
      se (melhormedia < media) entao
        melhormedia <- media
      fimse
    fimse
  fimpara
  mediageral <- mediageral / 10
  escreval("Média Geral = ", mediageral)
  escreval("Melhor Média = ", melhormedia)
  escreval("Pior Média = ", piormedia)
  escreval("-----")
finalgoritmo

```

18. Escrever um algoritmo que leia o valor de  $N$  e calcule os sucessivos valores de  $E$  usando a série abaixo e considerando os primeiros  $N$  termos, conforme exemplo com 5 termos a seguir:

$$E = 1/0! + 1/1! + 1/2! + 1/3! + 1/4! = 1 + 1 + 0.5 + 0.33 + 0.25 = 3.08$$

```

algoritmo "EX18"
var
  fat, i, j, n: inteiro
  expon: real
inicio
  escreva("Digite o número de termos: ")
  leia(n)
  expon <- 0

```



```

para i de 0 ate (n-1) passo 1 faca
    fat <- 1
    para j de i ate 1 passo -1 faca
        fat <- fat * j
    fimpara
    expon <- expon + 1/fat
    escreval("1/", i , "!")
fimpara
escreva("E = ", expon)
fimalgoritmo

```

19. Faça um algoritmo que leia um valor de entrada e monte uma pirâmide crescente, conforme exemplo a seguir:

**N=5**

```

1
1 2
1 2 3
1 2 3 4
1 2 3 4 5

```

```

algoritmo "EX19"
var
    n, i, j: inteiro
inicio
    escreva("Digite o valor de N: ")
    leia(n)
    para i de 1 ate n passo 1 faca
        para j de 1 ate i passo 1 faca
            escreva(j, " ")
        fimpara
        escreval()
    fimpara
fimalgoritmo

```

20. (Desafio) Dado um dia, mês e ano lido, monte o calendário daquele mês, conforme exemplo a seguir.

**30/10/2013**

Dom	Seg	Ter	Qua	Qui	Sex	Sab
-----						
		1	2	3	4	5
6	7	8	9	10	11	12
13	14	15	16	17	18	19
20	21	22	23	24	25	26
27	28	29	30	31		

Para saber o dia da semana, basta utilizar as fórmulas abaixo:

- a)  $a = (14 - \text{Mes}) \text{ div } 12$
- b)  $y = \text{Ano} - a$
- c)  $m = \text{Mes} + 12 * a - 2$
- d)  $q = \text{Dia} + (31 * m) \text{ div } 12 + y + y \text{ div } 4 - y \text{ div } 100 + y \text{ div } 400$
- e)  $d = q \text{ mod } 7$

E com o resultado final de  $d$ , verificar a seguir:

- $d = 0 \rightarrow$  Domingo
- $d = 1 \rightarrow$  Segunda-feira
- $d = 2 \rightarrow$  Terça-feira
- $d = 3 \rightarrow$  Quarta-feira
- $d = 4 \rightarrow$  Quinta-feira
- $d = 5 \rightarrow$  Sexta-feira
- $d = 6 \rightarrow$  Sábado

```
algoritmo "EX20"
var
    dia, mes, ano, a, y, m, q, d, sem, i: inteiro
inicio
    escreva("Digite o dia: ")
    leia(dia)
    escreva("Digite o mês: ")
    leia(mes)
    escreva("Digite o ano: ")
    leia(ano)
    escreval(dia, "/", mes, "/", ano)
    dia <- 1
    sem <- 0
    a <- (14 - mes) div 12
    y <- ano - a
    m <- mes + 12*a - 2
    q <- dia + (31*m) div 12 + y + y div 4 - y div 100 + y
div 400
    d <- q mod 7
    escreval("-----")
    escreval("Dom Seg Ter Qua Qui Sex Sáb")
    escreval("-----")
    para i de dia ate 31 passo 1 faca
        se (i = 1) entao
            enquanto (sem <> d) faca
                escreva(" ")
                sem <- sem + 1
            fimenquanto
        fimse
        se (i < 10) entao
            escreva(" ", i, " ")
        senao
```

```
        escreva(i, " ")
    fimse
    se (sem > 5) entao
        sem <- 0
        escreval()
    senao
        sem <- sem + 1
    fimse
fimpara
fimalgoritmo
```